

Cours: Approche à base de composants et concurrence

Filière: MR-GL2

Chapitre 4: la concurrence en Ada Partie 5/5: La gestion des rendez-vous

Réalisé par:

Dr. Sakka Rouis Taoufik

1

Ch 4: La concurrence en Ada (Partie 5/5)

I. Introduction

Rappelant que Ada est un langage de programmation fortement recommandé pour le développement des systèmes distribués.

En effet, le langage Ada offre des possibilités importantes vis-à-vis :

- **De la programmation structurée** : structuration de données (types prédéfinis et constructeurs de types simples et structurés) et structuration de traitements (structures de contrôle et les sous-programmes avec une distinction nette entre procédure et fonction). De plus, Ada est un langage fortement typé : il est plus sévère que Pascal!
- **De la programmation modulaire** en offrant un concept puissant, appelé package, doté de deux parties : interface (package) et implémentation (package body).

2

I. Introduction

- **De la programmation générique** en offrant la possibilité de concevoir et de réaliser des unités génériques : sous-programmes et paquetages. Ces unités peuvent être paramétrées sur de types, variables et sous-programmes.
- **De la gestion des exceptions**: ceci permet d'écrire des logiciels robustes dans divers domaines critiques : temps réel, systèmes embarqués.
- **De la traitement de la concurrence**: le langage Ada offre des constructions intéressantes permettant d'écrire des programmes concurrents fiables.
- **De l'analyse statique**: Ada est supporté par plusieurs outils permettant l'analyse statique d'un programme concurrent Ada tels que SPIN, SMV, INCA et FLAVERS. Sachant que l'analyse statique est une technique qui permet d'analyser un programme sans toutefois l'exécuter.

3

II. Les tâches

La concurrence en Ada est basée sur la synchronisation/communication des tâches.

Les tâches Ada permettent d'avoir des entités qui s'exécutent parallèlement et qui coopèrent selon les besoins pour résoudre les problèmes concurrents (ou non séquentiels).

Une tâche Ada (**task**) est une entité modulaire constituée de deux parties:

- **Une spécification** décrivant l'interface présentée aux autres tâches: cette partie inclus: le nom, Une partie visible avec déclarations d'entrées et Une partie privée avec déclarations d'entrées.

- **Un corps** décrivant son comportement dynamique. Cette partie inclus: le nom (voir spécification), une liste de déclarations et une liste d'instructions

4

II. Les tâches

Task Interface	<pre>task Server is entry take ; entry release ; end Server;</pre>
Task Body	<pre>task body Server is begin ... accept take ; ... accept release ; ... end Server;</pre>
Task Uses	<pre>task Client; task body Client is begin ... Server.take ; ... Server.release ; ... end Client;</pre>

5

II. Les tâches

Exemple 1: Soit trois activités indépendantes, on peut les exécuter : en séquence ou en parallèle

```
procedure Main is
begin
```

```
  Find_Ali_Phone_Nbr;
  Find_Slim_Phone_Nbr;
  Find_Md_Phone_Nbr;
```

```
end Main;
```

```
procedure Main is
```

```
  task Ali_Phone_Nbr;
```

```
  task Slim_Phone_Nbr;
```

```
  task body Ali_Phone_Nbr is
  begin
```

```
    Find_Ali_Phone_Nbr;
```

```
  end Ali_Phone_Nbr;
```

```
  task body Slim_Phone_Nbr is
  begin
```

```
    Find_Slim_Phone_Nbr;
```

```
  end Slim_Phone_Nbr;
```

```
begin
```

```
  Find_Md_Phone_Nbr;
```

```
end Main;
```

6

II. Les tâches

Exemple 2: Soit deux activités indépendantes T1 et T2, on peut les exécuter en parallèle comme suite:

```
with Text_io;
use Text_io;

procedure Ord is
  task T1;
  task T2;

  task body T1 is
  begin
    loop
      Put("T1");
    end loop;
  end T1;

  task body T2 is
  begin
    loop
      Put("T2");
    end loop;
  end T2;

  begin
    null ;
  end Ord;
```

7

II. Les tâches

L'activation d'une tâche est automatique.

Les tâches T1 et T2 deviennent actives quand l'unité parente atteint le **begin** qui suit la déclaration des tâches.

Ainsi, ces deux tâches, dépendantes, sont activées en parallèle avec la tâche principale, qui est le programme principal.

La tâche principale attend que les tâches dépendantes s'achèvent pour terminer. La terminaison comporte donc deux étapes : elle s'achève quand elle atteint le **end** final, puis elle ne deviendra terminée que lorsque les tâches dépendantes, s'il y en a, seront terminées.

8

III. Les rendez-vous

Les tâches Ada peuvent **interagir entre elles** (se **communiquer entre elles**) grâce à des entrées de tâche. Ce mécanisme est appelé **rendez-vous**.

Une tâche publie ses entrées et leurs signatures dans sa spécification. Une entrée d'une tâche correspond à une procédure appellable par une tâche appelante.

L'appelante invoque une entrée de l'appelée **et se bloque** en attendant que l'appelée accepte

L'appelée accepte les appels sur ses entrées **et se bloque** en absence de demandes

Lorsque appelante et appelée sont prêtes, l'appel (le rendez-vous) se déroule dans le contexte de l'appelée

III. Les rendez-vous

```
with Text_IO;
use Text_IO;
procedure Ecran is
task T1;
task T2;
task Allocateur is
  entry Prendre;
  entry Relacher;
end Allocateur;
task body Allocateur is
begin
  loop
    select
      accept Prendre;
      accept Relacher;
    or
      terminate;
    end select;
  end loop;
end Allocateur;
```

```
task body T1 is begin
  Allocateur.Prendre;
  for I in 1 .. 3 loop
    Put("T1");
  end loop;
  Allocateur.Relacher;
end T1;
```

```
task body T2 is begin
  Allocateur.Prendre;
  for I in 1 .. 3 loop
    Put("T2");
  end loop ;
  Allocateur.Relacher;
end T2;
```

```
begin
  null;
end Ecran;
```

Ce programme
affiche aléatoirement
T1T1T1T2T2T2 ou
T2T2T2T1T1T1

III. Les rendez-vous

```

with Text_IO;
use Text_IO;
procedure Ecran is
task T1;
task T2;
task Allocateur is
  entry Prendre;
  entry Relacher;
end Allocateur;
task body Allocateur is
begin
  loop
    select
      accept Prendre;
      accept Relacher;
    or
      terminate;
    end select;
  end loop;
end Allocateur;

```

```

task body T1 is begin
  loop
    Allocateur.Prendre;
    for I in 1 .. 3 loop
      Put("T1");
    end loop;
    Allocateur.Relacher;
  end loop ;
end T1;
task body T2 is begin
  loop
    Allocateur.Prendre;
    for I in 1 .. 3 loop
      Put("T2");
    end loop ;
    Allocateur.Relacher;
  end loop ;
end T2;
begin
  null;
end Ecran;

```

Ce programme
affiche aléatoirement
une combinaison de
séquence de 3 T1 et
de 3 T2

Exp:
T1T1T1T2T2T2T1T1
T1T2T2T2...

11

III. Les rendez-vous

On peut conclure que sur le plan conceptuel, les tâches peuvent être classées en trois catégories :

- **Les tâches serveuses** qui proposent des rendez-vous et n'en demandent pas : par exemple la tâche Allocateur.
- **Les tâches actrices** qui ne proposent pas des rendez-vous. Mais elles en demandent : par exemple les tâches T1 et T2.
- **Les tâches actrices/serveuses** qui proposent et demandent des rendez-vous.

12

III. Les rendez-vous

Le mécanisme de rendez-vous Ada permet à la fois **la synchronisation et la communication**, nécessaires à la résolution des problèmes de compétition et de coopération entre tâches.

Une tâche Ada ne peut traiter qu'un seul rendez-vous à la fois, c'est le principe **d'exclusion mutuelle** sur les rendez-vous.

Chaque entrée proposée par une tâche serveuse est dotée d'une file d'attente FIFO (Structure de Données First In First Out) gérée par l'exécutif d'Ada. Cette file FIFO permet de stoker les demandes sur cette entrée dans l'ordre d'arrivée.

13

III. Les rendez-vous

Pour gérer le non déterminisme, le langage Ada dispose d'une instruction adéquate appelée l'instruction d'attente sélective **select**.

Cette l'instruction **select** permet à la tâche appelée de sélectionner **arbitrairement** une demande de rendez-vous **parmi les rendez-vous disponibles**.

```
select
    accept entree_A;    ...
or
    accept entree_B;    ...
or
    accept entree_C; ...
end select;
```

14

III. Les rendez-vous

La durée du rendez-vous est la durée nécessaire à l'exécution des instructions attachées à l'entrée :

```
accept nom_entree do
    <Instructions exécutées>
end nom_entree ;
```

La communication entre tâches est réalisée par la transmission de paramètres typés dans les deux sens (in, out et in out), lors de l'occurrence d'un rendez-vous.

Exemple: task Stream is
 public entry input (data: in String) ;
 public entry output (data: out String) ;
end Stream;

L'exemple suivant illustre l'utilisation exclusive d'une ressource critique (variable entière encapsulée par la tâche serveuse **variable_protegee**) par deux tâches actrices t1 et t2.

15

III. Les rendez-vous

```
procedure LireEcrire is
task variable_protegee is
    entry lire (v : out integer);
    entry ecrire (v : in integer);
end variable_protegee ;
task body variable_protegee is
    x : integer := 0;
begin
loop
select
    accept lire (v : out integer) do
        v := x ;
        Put("c'est une lecture à partir de la variable protégée");
    end lire ;
    or
    accept ecrire (v : in integer) do
        x := v ;
        Put("C'est une écriture dans la variable protégée");
    end ecrire ;
    or
    terminate ;
end select ;
end loop ;
end variable_protegee ;
```

```
task t1 ;
task t2 ;
task body t1 is
    a: integer:=0;
begin
    ...
    variable_protegee.ecrire (a);
    ...
end t1 ;

task body t2 is
    b: integer;
begin
    ...
    variable_protegee.lire(b);
    ...
end t2 ;
begin
    null;
end LireEcrire ;
```

16

III. Les rendez-vous


L'exemple précédent illustre deux cas.

- **Dans le cas a:** la tâche APPELEE est prête après que la tâche APPELANTE ait effectué une demande de rendez-vous sur une entrée.
- **Dans le cas b:** c'est la tâche APPELEE qui est prête avant la demande de rendez-vous.

Lors d'un appel, si la tâche APPELEE attend sur l'instruction *accept*, le rendez-vous a lieu immédiatement, sinon, la demande de la tâche APPELANTE est mémorisée dans une file FIFO et la tâche est mise en attente passive de l'acceptation du rendez-vous. Ainsi, la forme de communication entre les tâches Ada est donc **bloquante**.

17

III. Les rendez-vous



18

IV. Les types tâches

A l'instar des types de données usuels ou des types abstraits, il est possible de déclarer des types de tâches (task type).

Les types tâches en Ada sont considérés comme **limited private** : seule la transmission est permise.

Les affectations et les comparaisons des tâches ne sont pas permises.

Les types tâches permettent de factoriser plusieurs tâches ayant le même comportement. De plus, ils ouvrent la perspective de créer dynamiquement des tâches.

19

IV. Les types tâches

-- type de tâche et non une tâche

```
task type allocateur is
  entry prendre;
  entry liberer;
end allocateur;
task body allocateur is
begin
  loop
    select
      accept prendre;
      accept liberer;
    or
      terminate;
    end select;
  end loop;
end allocateur;
```

-- définition statique des tâches
ecran : allocateur;
imprimante : allocateur;

Les deux tâches serveuses *ecran* et *imprimante* permettent l'accès exclusif (par des tâches actrices T1, T2, ...) respectivement aux deux ressources critiques écran et imprimante.

20